



Techniques d'intégration (partie 3 de 5)

Substitutions trigonométriques

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

Ce document est une révision de celui produit en 2003. L'objectif principal de cette feuille Maple est de rendre l'élève apte à transposer en Maple la technique d'intégration par substitution trigonométrique. Cette transposition sera réalisée avec certaines macro-commandes de la sous-bibliothèque [Calculus1](#) de la bibliothèque [Student](#).


Les exemples développés dans ce document illustreront la manière d'obtenir l'intégrale indéfinie de deux façons:

- une première façon consistera à obtenir directement l'intégrale indéfinie avec la macro-commande [int](#) de la bibliothèque principale, question d'observer le mécanisme de la simplification automatique de Maple.
- la seconde façon, quant à elle, consistera à faire un développement pas.à pas. Un tel développement sera réalisé selon les règles d'intégration [Rule](#) de *Student*[*Calculus1*].

Pour compléter tout développement pas à pas, la dernière étape consistera à faire une vérification de la réponse finale par dérivation.

Attention:

- S'assurer que la rubrique « Afficher les étiquettes des équations » soit cochée: voir *menu Outils* → *Options...* → *Affichage*.
- S'assurer que le préfixe de numérotation d'étiquettes soit initialisée à 1 avec un schéma de numérotation séquentielle: voir *menu Format* → *Étiquettes d'équation (L)* → *Affichage des étiquettes... (L)*.



- L'insertion d'une étiquette est obtenu avec le raccourci « ctrl + 1 » (1 pour *label* en anglais).

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
with(Student[Calculus1],Rule,Integrand);
infolevel[Student[Calculus1]]:=1:
                                     [Rule,Integrand]
```

(1)

Assurance que les étiquettes des équations soient affichées.

```
> interface(labelling=true)
```

true

(2)

Réglage de l'affichage pour les variables conditionnées.

```
> with(Typesetting):
interface(typesetting=standard); #niveau de composition Maple
Standard
interface(showassumed=2);      # Variables avec suppositiosnOhrase
                                extended, [extended]
                                2
```

(3)

Intégration par substitution trigonométrique

Lorsque l'intégrande renferme la forme $\sqrt{a^2 - x^2}$ ($a > 0$) on pose habituellement $x = a \sin(\theta)$ avec $dx = a \cos(\theta) d\theta$. En conséquence $\sqrt{a^2 - x^2} = \sqrt{a^2 (1 - \cos^2(\theta))} = \sqrt{a^2 (\sin^2(\theta))} = a |\sin(\theta)|$ ($a > 0$).

Le domaine de définition de $\sqrt{a^2 - x^2}$ ($a > 0$) étant $x \in [-a, a]$, il faut que $-a \leq a \sin(\theta) \leq a$, soit $-1 \leq \sin(\theta) \leq 1$. C'est le cas lorsque $\theta \in \left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$ et donc, dans cet intervalle $\cos(\theta) \geq 0$.

Ainsi, pour $\theta \in \left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$, $\sqrt{a^2 - x^2} \Big|_{x = a \sin(\theta)} \Rightarrow \sqrt{a^2 - x^2} = a \cos(\theta)$ et, pour la substitution à rebours, $\theta = \arcsin\left(\frac{x}{a}\right)$.

Lorsque l'intégrande renferme la forme $\sqrt{x^2 - a^2}$ ($a > 0$) on pose habituellement $x = a \sec(\theta)$ avec $dx = a \sec(\theta) \tan(\theta) d\theta$. En conséquence $\sqrt{x^2 - a^2} = \sqrt{a^2 (\sec^2(\theta) - 1)} = \sqrt{a^2 (\tan^2(\theta))} = a |\tan(\theta)|$ ($a > 0$).

Le domaine de définition de $\sqrt{x^2 - a^2}$ ($a > 0$) étant $x \in]-\infty, -a] \cup [a, \infty[$, il faut que $a \sec(\theta) \leq -a$ ou $a \sec(\theta) \geq a$. Soit $\sec(\theta) \leq -1$ ou $\sec(\theta) \geq 1$.

C'est le cas lorsque $\theta \in \left[0, \frac{\pi}{2}\right[\cup \left] \frac{\pi}{2}, \pi\right]$.

Ainsi,

• pour $\theta \in \left[0, \frac{\pi}{2}\right[$, $\sqrt{x^2 - a^2} \Big|_{x = a \sec(\theta)} \Rightarrow \sqrt{x^2 - a^2} = a \tan(\theta)$ et, pour la substitution à rebours,

$\theta = \arccos\left(\frac{a}{x}\right)$.

• pour $\theta \in \left] \frac{\pi}{2}, \pi\right]$, $\sqrt{x^2 - a^2} \Big|_{x = a \sec(\theta)} \Rightarrow \sqrt{x^2 - a^2} = -a \tan(\theta)$ et, pour la substitution à rebours,

$$\theta = \arccos\left(\frac{a}{x}\right).$$

Avec la substitution trigonométrique $x = a \sec(\theta)$, il faudra donc être très circonspect quant au mécanisme de la simplification automatique de Maple. En effet, pour obtenir une simplification appropriée, il est nécessaire de conditionner la variable θ , sinon cela pourrait causer quelques problèmes de développement dans \mathbb{R} .

Simplification conditionnelle de $x = a \sec(\theta)$

$$\begin{aligned} > \text{theta} := 'theta'; \\ & \qquad \qquad \qquad \theta := \theta \end{aligned} \tag{4}$$

$$\begin{aligned} > \text{Formule} := \text{sqrt}(x^2 - a^2); \\ & \qquad \qquad \qquad \text{Formule} := \sqrt{-a^2 + x^2} \end{aligned} \tag{5}$$

$$\begin{aligned} > \text{subs}(x = a * \sec(\text{theta}), \text{Formule}); \\ & \qquad \qquad \qquad \sqrt{-a^2 + a^2 \sec(\theta)^2} \end{aligned} \tag{6}$$

$$\begin{aligned} > \text{simplify}((6)) \text{ assuming}(a > 0); \\ & \qquad \qquad \qquad a \sqrt{\frac{\sin(\theta)^2}{\cos(\theta)^2}} \end{aligned} \tag{7}$$

$$\begin{aligned} > \text{simplify}((7)) \text{ assuming}(\text{theta}, \text{RealRange}(0, \text{Open}(\text{Pi}/2))); \\ & \qquad \qquad \qquad \frac{a \sin(\theta)}{\cos(\theta)} \end{aligned} \tag{8}$$

$$\begin{aligned} > \text{simplify}((7)) \text{ assuming}(\text{theta}, \text{RealRange}(\text{Open}(\text{Pi}/2), \text{Pi})); \\ & \qquad \qquad \qquad -\frac{a \sin(\theta)}{\cos(\theta)} \end{aligned} \tag{9}$$

Lorsque l'intégrande renferme la forme $\sqrt{a^2 + x^2}$ ($a > 0$) on pose habituellement $x = a \tan(\theta)$ avec $dx = a \sec^2(\theta) d\theta$. En conséquence $\sqrt{a^2 + x^2} = \sqrt{a^2 (1 + \tan^2(\theta))} = \sqrt{a^2 (\sec^2)(\theta)} = a |\sec \theta|$ ($a > 0$).

Le domaine de définition de $\sqrt{a^2 + x^2}$ ($a > 0$) étant \mathbb{R} , il faut que $a \tan(\theta) \in \mathbb{R}$. c'est-à-dire que $\tan(\theta) \in \mathbb{R}$. C'est le cas lorsque $\theta \in \left] -\frac{\pi}{2}, \frac{\pi}{2} \right[$.

Ainsi, pour $\theta \in \left] -\frac{\pi}{2}, \frac{\pi}{2} \right[$, $\left. \sqrt{a^2 + x^2} \right|_{x = a \tan(\theta)} \Rightarrow \sqrt{a^2 + x^2} = a \sec(\theta)$ et, pour la substitution à

rebours, $\theta = \arctan\left(\frac{x}{a}\right)$.

Note: On pourra faire cette substitution lorsqu'on a une forme $a^2 + x^2$ sans radical dans l'intégrande.

Par contre, en posant la substitution algébrique $x = -\frac{a^2 + u^2}{2u}$ ($a > 0$) avec $dx = \frac{a^2 - u^2}{2u^2} du$.

Ainsi, puisque $x = -\frac{a^2 + u^2}{2u}$ est définie pour $u \in \mathbb{R} \setminus \{0\}$, x l'est en particulier pour $u \in]-a, a[$. D'autre part $u \rightarrow -\frac{a^2 + u^2}{2u}$ est injective. ainsi

$$\bullet u \in]-a, a[\left| \begin{array}{l} \sqrt{x^2 - a^2} \\ x = -\frac{a^2 + u^2}{2u} \end{array} \right. \Rightarrow \sqrt{x^2 - a^2} = \sqrt{-a^2 + \frac{(a^2 + u^2)^2}{4u}} =$$

La technique d'intégration par substitution trigonométrique est en fait une technique d'intégration par changement de variables. Les règles d'intégration *Rule* de la bibliothèque *Student[CalculusI]* serviront donc à réaliser une substitution trigonométrique dans une intégrale indéfinie (Voir document Maple *Changement de variables*). Cette technique de changement de variables consiste à reformuler la différentielle de départ en termes de fonctions trigonométriques plus facile à intégrer.

En plus d'aborder la technique d'intégration par substitution trigonométrique, ce document-ci sera l'occasion d'expérimenter un nombre accru de macro-commandes dédiées à la simplification sur demande. À ce niveau, la transposition dans Maple de la technique d'intégration par substitution trigonométrique exige un niveau de compétence assez élevé dans la simplification sur demande. En plus de la connaissance des macro-commandes habituelles dédiées à la simplification sur demande, la connaissance de plusieurs identités trigonométriques de base est nécessaire.

Exemple 1

Calculons l'intégrale indéfinie $\int \frac{x^2}{\sqrt{36 - x^2}} dx$.

Question d'observer le mécanisme de la simplification automatique, commençons par obtenir directement

avec la macro-commande *int*, l'intégrale indéfinie $\int \frac{x^2}{\sqrt{36 - x^2}} dx$.

Pour bien documenter les prochains développements, créons d'abord une fonction *f* (avec l'opérateur flèche) dont la règle *f(x)* sera l'intégrande.

```
> f:=x->x^2/sqrt(36-x^2);
```

$$f := x \rightarrow \frac{x^2}{\sqrt{36 - x^2}} \quad (10)$$

Pour que l'intégrale indéfinie puisse s'afficher dans la zone des résultats avec une notation mathématique habituelle, utilisons la forme inactive de la macro-commande *int*. De plus, donnons le nom *Problème* à l'intégrale indéfinie.

```
> Problème:=Int(f(x),x);
```

$$\text{Problème} := \int \frac{x^2}{\sqrt{-x^2 + 36}} dx \quad (11)$$

Obtenons directement cette intégrale indéfinie avec la macro-commande *value*. Assignons le résultat au nom *primitive*.

```
> primitive:=value(Problème);
```

$$\text{primitive} := -\frac{1}{2} x \sqrt{-x^2 + 36} + 18 \arcsin\left(\frac{1}{6} x\right) \quad (12)$$

Rappelons que pour obtenir l'ensemble de toutes les primitives, il faut gérer manuellement la globalisation des constantes d'intégration. Ajoutons donc une constante *C* d'intégration.

```
> Rép_Maple:=Problème=primitive+C;
```

$$\text{Rép_Maple} := \int \frac{x^2}{\sqrt{-x^2 + 36}} dx = -\frac{1}{2} x \sqrt{-x^2 + 36} + 18 \arcsin\left(\frac{1}{6} x\right) + C \quad (13)$$

Intégrons maintenant en effectuant un développement pas à pas jusqu'aux modèles de base. Rappelons-nous l'énoncé de l'intégrale indéfinie à calculer.

```
> Problème;
```

$$\int \frac{x^2}{\sqrt{-x^2 + 36}} dx \quad (14)$$

Intégrons $\int \frac{x^2}{\sqrt{36 - x^2}} dx$ par substitution trigonométrique. Posons $x = 6 \sin(\theta)$ avec $\theta \in \left[-\frac{\pi}{2}, \frac{\pi}{2}\right]$

```
> Rule[change,x=6*sin(theta),theta]((14));
```

```
Creating problem #1
```

```
Applying substitution x = 6*sin(theta), theta = arcsin(1/6*x) with dx = 6*cos(theta)*dtheta, dtheta = 1/(-x^2+36)^(1/2)*dx
```

$$\int \frac{x^2}{\sqrt{-x^2 + 36}} dx = \int 36 \sin^2(\theta) d\theta \quad (15)$$

Nous avons à intégrer une puissance paire de sinus. Réécrivons (14) à l'aide de l'identité

$$\sin^2(\theta) = \frac{1 - \cos(2\theta)}{2}$$

```
> Rule[rewrite,sin(theta)^2 = (1-cos(2*theta))*(1/2)]((15));
```

$$\int \frac{x^2}{\sqrt{-x^2 + 36}} dx = \int (-18 \cos(2\theta) + 18) d\theta \quad (16)$$

```
> Rule[sum]((16));
```

$$\int \frac{x^2}{\sqrt{-x^2 + 36}} dx = \int (-18 \cos(2\theta)) d\theta + \int 18 d\theta \quad (17)$$

```
> Rule[change,u=2*theta]((17));
```

Applying substitution $\theta = 1/2*u$, $u = 2*\theta$ with $d\theta = 1/2*du$,
 $du = 2*d\theta$

$$\int \frac{x^2}{\sqrt{-x^2+36}} dx = \int (-9 \cos(u)) du + \int 18 d\theta \quad (18)$$

```
> Rule[`c*`](18);
```

$$\int \frac{x^2}{\sqrt{-x^2+36}} dx = -9 \left(\int \cos(u) du \right) + \int 18 d\theta \quad (19)$$

```
> Rule[cos](19);
```

$$\int \frac{x^2}{\sqrt{-x^2+36}} dx = -9 \sin(u) + \int 18 d\theta \quad (20)$$

```
> Rule[constant](20);
```

$$\int \frac{x^2}{\sqrt{-x^2+36}} dx = -9 \sin(u) + 18 \theta \quad (21)$$

```
> Rule[revert](21);
```

Reverting substitution using $u = 2*\theta$

$$\int \frac{x^2}{\sqrt{-x^2+36}} dx = -9 \sin(2\theta) + 18 \theta \quad (22)$$

```
> Rule[revert](22);
```

Reverting substitution using $\theta = \arcsin(1/6*x)$

$$\int \frac{x^2}{\sqrt{-x^2+36}} dx = -\frac{1}{2} x \sqrt{-x^2+36} + 18 \arcsin\left(\frac{1}{6} x\right) \quad (23)$$

```
> Rép_finale:=(23)+(0=C);
```

$$\text{Rép_finale} := \int \frac{x^2}{\sqrt{-x^2+36}} dx = -\frac{1}{2} x \sqrt{-x^2+36} + 18 \arcsin\left(\frac{1}{6} x\right) + C \quad (24)$$

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

(25)

$$\frac{\partial}{\partial x} \left(-\frac{1}{2} x \sqrt{-x^2 + 36} + 18 \arcsin\left(\frac{1}{6} x\right) + C \right) = -\frac{1}{2} \sqrt{-x^2 + 36} + \frac{1}{2} \frac{x^2}{\sqrt{-x^2 + 36}} + \frac{18}{\sqrt{-x^2 + 36}} \quad (25)$$

Simplifions le membre de droite.

```
> ``=normal(rhs((25)));
```

$$= \frac{x^2}{\sqrt{-x^2 + 36}} \quad (26)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$\frac{x^2}{\sqrt{-x^2 + 36}} \quad (27)$$

Exemple 2

Calculons l'intégrale indéfinie $\int \frac{e^x}{4e^{2x} + 9} dx$.

Remarque: Cette intégrale indéfinie est celle de l'exemple 6 du document « Techniques d'intégration par changement de variables (partie 1 de 5) ».

Avec Maple directement.

```
> f:=x->exp(x)/(4*exp(2*x)+9);
```

$$f := x \rightarrow \frac{e^x}{4e^{2x} + 9} \quad (28)$$

Posons l'intégrale indéfinie à calculer.

```
> Problème:=Int(f(x),x);
```

$$\text{Problème} := \int \frac{e^x}{4e^{2x} + 9} dx \quad (29)$$

Obtenons maintenant cette intégrale indéfinie avec la macro-commande *value*. Assignons le résultat au nom primitive.

```
> primitive:=value(Problème);
```

$$\text{primitive} := \frac{1}{6} \arctan\left(\frac{2}{3} e^x\right) \quad (30)$$

```
> Rép_Maple:=Problème=primitive+C;
```

$$\text{Rép_Maple} := \int \frac{e^x}{4e^{2x} + 9} dx = \frac{1}{6} \arctan\left(\frac{2}{3} e^x\right) + C \quad (31)$$

Maintenant, avec un développement pas à pas jusqu'aux modèles de base. Rappelons-nous l'énoncé de l'intégrale indéfinie qu'on cherche à obtenir.

```
> Problème;
```

$$\int \frac{e^x}{4e^{2x} + 9} dx \quad (32)$$

Le dénominateur de l'intégrande est de la forme $a^2 + x^2$. Posons la substitution trigonométrique

$$e^x = \frac{3 \tan(\theta)}{2} \text{ où } \theta \in \left] 0, \frac{\pi}{2} \right[\text{ (car } e^x > 0)$$

```
> Rule[change,exp(x) = 3*tan(theta)*(1/2)]((32));
```

```
Creating problem #2
```

```
Applying substitution x = ln(3)-ln(2)+ln(tan(theta)), theta = arctan
(2/3*exp(x)) with dx = (1+tan(theta)^2)/tan(theta)*dtheta, dtheta = 2/3*
exp(x)/(4/9*exp(x)^2+1)*dx
```

$$\int \frac{e^x}{4e^{2x} + 9} dx = \int \frac{1}{6} d\theta \quad (33)$$

```
> Rule[constant]((33));
```

$$\int \frac{e^x}{4e^{2x} + 9} dx = \frac{1}{6} \theta \quad (34)$$

Ayant posé $e^x = \frac{3 \tan(\theta)}{2}$, effectuons à rebours la substitution $\theta = \arctan\left(\frac{2e^x}{3}\right)$ et ajoutons manuellement une constante d'intégration.

```
> Rule[revert]((34));
```

```
Reverting substitution using theta = arctan(2/3*exp(x))
```

$$\int \frac{e^x}{4e^{2x} + 9} dx = \frac{1}{6} \arctan\left(\frac{2}{3} e^x\right) \quad (35)$$

```
> Rép_finale := (35) + (0=C);
```

$$\text{Rép_finale} := \int \frac{e^x}{4e^{2x} + 9} dx = \frac{1}{6} \arctan\left(\frac{2}{3} e^x\right) + C \quad (36)$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x} \left(\frac{1}{6} \arctan\left(\frac{2}{3} e^x\right) + C \right) = \frac{1}{9} \frac{e^x}{\frac{4}{9} (e^x)^2 + 1} \quad (37)$$

```
> ``=simplify(rhs((37)));
```

$$= \frac{e^x}{4e^{2x} + 9} \quad (38)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$\frac{e^x}{4e^{2x} + 9} \quad (39)$$

Exemple 3

Calculons l'intégrale indéfinie $\int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx$.

Remarque: Le calcul de cette intégrale indéfinie présente quelques difficultés de simplification. Mais, le développement pas à pas est enrichissant à plusieurs points de vue. En particulier, cet exemple permet de constater que le mécanisme de simplification automatique de Maple n'est pas adapté pour l'analyse dans les réels. Tous les calculs de Maple sont faits sur le corps des nombres complexes. Même lorsqu'on utilise les requêtes de la sous-bibliothèque *Calculus1* (dont les développements sont, en principe, faits sur le corps des réels) le mécanisme de la simplification automatique ne donne pas, parfois, un résultat attendu dans les nombres réels. En particulier, l'absence de l'utilisation de la valeur absolue dans le mécanisme de la simplification de Maple est approprié par l'analyse complexe mais nous est problématique en analyse réelle.

Voici quelques exemples illustrant la problématique que pose dans \mathbb{R} les règles du mécanisme de la simplification automatique.

Soit la fonction f définie par $f(x) = \frac{\sqrt{x+1}}{\sqrt{x-1}}$. Il est clair que le domaine de f est $]1, \infty[$

```
> f:=x->sqrt(x+1)/sqrt(x-1);
```

$$f := x \rightarrow \frac{\sqrt{x+1}}{\sqrt{x-1}} \quad (40)$$

Observez la simplification automatique de $f(4)$.

```
> f(-4);
```

$$\frac{1}{5} \sqrt{3} \sqrt{5} \quad (41)$$

Par contre, le macro-commande *eval* appelle une substitution sans simplification automatique dans \mathbb{C} ... presque car il y a eu rationalisation automatique du dénominateur dans le cas de valeurs numériques.


```
> eval(f(x), x=-4);
```

$$-\frac{1}{5} \sqrt{-3} \sqrt{-5} \quad (42)$$

Bien sûr, la réduction du résultat imaginaire (42) est un nombre réel.

La macro-commande *plot* appelle aussi une simplification automatique dans \mathbb{C} .


```
> plot([x, f(x), x=-4..4]);
```



Evidemment, dans \mathbb{R} , il ne peut y avoir de tracé pour $x < 1$.

Même l'utilisation d'un environnement réel ne bloque en rien la réduction (simplification) automatique dans \mathbb{C} .

```
> use RealDomain in plot([x,f(x), x=-4..4]) end use;
```



Autre exemple. On connaît bien le modèle de base $\int \frac{1}{x} dx = \ln(|x|) + C$.

```
> Int(1/x,x)=int(1/x,x);
```

$$\int \frac{1}{x} dx = \ln(x)$$

(43)

Ce n'est pas une erreur... pour Maple. On est dans \mathbb{C} et non pas dans \mathbb{R} .

$$\begin{aligned} & \left[\begin{array}{l} > \text{Int}(\sec(\theta), \theta) = \text{int}(\sec(\theta), \theta); \\ \int \sec(\theta) \, d\theta = \ln(\sec(\theta) + \tan(\theta)) \end{array} \right. \end{aligned} \quad (44)$$

Ce qui ne correspond pas au modèle vue en classe (dans \mathbb{R}).

Il arrive tout de même parfois que Maple nous rappelle qu'on est dans \mathbb{C} . On connaît bien l'identité

$$\begin{aligned} & \sqrt{x^2} = |x|. \\ & \left[\begin{array}{l} > \mathbf{x := 'x':} \\ \quad \text{simplify}(\text{sqrt}(x^2)); \\ \qquad \qquad \qquad \text{csgn}(x) x \end{array} \right. \end{aligned} \quad (45)$$

Oups!

$$\left[\begin{array}{l} > \text{use RealDomain in simplify}(\text{sqrt}(x^2)) \text{ end use;} \\ \qquad \qquad \qquad |x| \end{array} \right. \quad (46)$$

$$\left[\begin{array}{l} > (-8)^{(1/3)}; \\ \qquad \qquad \qquad (-8)^{1/3} \end{array} \right. \quad (47)$$

L'écriture $(-8)^{\frac{1}{3}}$ correspond à la racine cubique principale de -8 qui est une valeur *imaginaire*. Par contre, dans \mathbb{R} , $(-8)^{\frac{1}{3}} = -2$

$$\left[\begin{array}{l} > \text{use RealDomain in } (-8)^{(1/3)} \text{ end use;} \\ \qquad \qquad \qquad -2 \end{array} \right. \quad (48)$$

Intégrons maintenant directement $\int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} \, dx$, question d'observer la simplification automatique de

Maple.

$$\left[\begin{array}{l} > \mathbf{f := x -> (3*x^2 - 4) / \text{sqrt}(x^2 - 9);} \\ \qquad \qquad \qquad f := x \rightarrow \frac{3x^2 - 4}{\sqrt{x^2 - 9}} \end{array} \right. \quad (49)$$

Donnons le nom *Problème* à l'intégrale indéfinie qu'il faut obtenir.

$$\left[\begin{array}{l} > \mathbf{Problème := Int(f(x), x);} \\ \qquad \qquad \qquad \text{Problème} := \int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} \, dx \end{array} \right. \quad (50)$$

Obtenons maintenant cette intégrale indéfinie avec la macro-commande *value*. Assignons le résultat au nom *primitive*.

$$\left[\begin{array}{l} > \mathbf{primitive := value(Problème);} \\ \qquad \qquad \qquad \text{primitive} := \frac{3}{2} x \sqrt{x^2 - 9} + \frac{19}{2} \ln(x + \sqrt{x^2 - 9}) \end{array} \right. \quad (51)$$

> `Rép_Maple:=Problème=primitive+C;`

$$\text{Rép_Maple} := \int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx = \frac{3}{2} x \sqrt{x^2 - 9} + \frac{19}{2} \ln(x + \sqrt{x^2 - 9}) + C \quad (52)$$

ATTENTION: Il revient toujours à l'utilisateur de surveiller le mécanisme de simplification automatique. Sur le domaine (réel) de définition de l'intégrande, ce résultat n'est pas valide. En effet, le domaine de définition de l'intégrande est $\mathbb{R} \setminus [-3, 3]$ et on remarque facilement que $\frac{19 \ln(x + \sqrt{x^2 - 9})}{2}$ n'est pas réel pour $x < -3$.

Bref, sûrement que la primitive obtenue directement (51) en est une pour seulement $x > 3$. Qu'en est-il pour $x < -3$?

Maintenant, allons-y avec un développement pas à pas jusqu'aux modèles de base. Rappelons-nous l'énoncé de l'intégrale indéfinie qu'on cherche à obtenir.

> `Problème;`

$$\int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx \quad (53)$$

Effectuons une substitution trigonométrique en posant $x = 3 \sec(\theta)$ où $\theta \in \left[0, \frac{\pi}{2} \right] \cup \left[\frac{\pi}{2}, \pi\right]$.

> `Rule[change,x=3*sec(theta)]((53));`

Creating problem #3

Applying substitution $x = 3 \sec(\theta)$, $\theta = \text{arcsec}(1/3*x)$ with $dx = 3 \sec(\theta) \tan(\theta) d\theta$, $d\theta = 3/x^2 / (1-9/x^2)^{(1/2)} dx$

$$\int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx = \int (27 \sec^3(\theta) - 4 \sec(\theta)) d\theta \quad (54)$$

Nous avons vu que la substitution $x = a \sec(\theta)$ nous amène à une simplification conditionnelle de $\sqrt{x^2 - a^2}$. Ce qui n'a pas été pris en compte dans la substitution précédente. Nous devons donc faire un développement pas à pas autrement qu'avec les règles de *Student[Calculus1]*.

Effectuons donc manuellement la simplification conditionnelle cette substitution trigonométrique.

> `intégrandetheta1:=eval(Integrand(Problème),x=3*sec(theta))[1];`

$$\text{intégrandetheta1} := \frac{27 \sec^3(\theta) - 4}{\sqrt{9 \sec^2(\theta) - 9}} \quad (55)$$

Calcul de la différentielle dtheta1.

> `dthata1:=diff(3*sec(theta),theta);`

$$d\theta_1 := 3 \sec(\theta) \tan(\theta) \quad (56)$$

Nouvelle intégration.

> `Int(intégrandetheta1*dthata1,theta);`

(57)

$$\int \frac{3(27 \sec(\theta)^2 - 4) \sec(\theta) \tan(\theta)}{\sqrt{9 \sec(\theta)^2 - 9}} d\theta \quad (57)$$

```
> algsubs(sec(theta)^2-1=tan(theta)^2,(57));
```

$$\int \frac{1}{3} \frac{\sqrt{9} \tan(\theta) \sec(\theta) (27 \tan(\theta)^2 + 23)}{\sqrt{\tan(\theta)^2}} d\theta \quad (58)$$

Simplification conditionnelle pour $\theta \in \left] 0, \frac{\pi}{2} \right[$.

```
> Intégrale_1:=simplify((58)) assuming(cos(theta)>0, sin(theta)>0);
```

$$\text{Intégrale}_1 := - \left(\int \frac{4 \cos(\theta)^2 - 27}{\cos(\theta)^3} d\theta \right) \quad (59)$$

Simplification conditionnelle pour $\theta \in \left] \frac{\pi}{2}, \pi \right[$.

```
> Intégrale_2:=simplify((58)) assuming(cos(theta)<0, sin(theta)>0);
```

$$\text{Intégrale}_2 := \int \frac{4 \cos(\theta)^2 - 27}{\cos(\theta)^3} d\theta \quad (60)$$

Trouvons Intégrale_1.

```
> Intégrale_1=expand(Intégrale_1);
```

$$- \left(\int \frac{4 \cos(\theta)^2 - 27}{\cos(\theta)^3} d\theta \right) = - \left(\int \left(\frac{4}{\cos(\theta)} - \frac{27}{\cos(\theta)^3} \right) d\theta \right) \quad (61)$$

```
> ``=algsubs(cos(theta)^(-1)=sec(theta),rhs((61)));
```

$$= - \left(\int (-27 \sec(\theta)^3 + 4 \sec(\theta)) d\theta \right) \quad (62)$$

```
> ``=expand(rhs((62)));
```

$$= 27 \left(\int \sec(\theta)^3 d\theta \right) - 4 \left(\int \sec(\theta) d\theta \right) \quad (63)$$

On reconnaît le modèle $\int \sec(\theta) d\theta$, par contre, $\int \sec(\theta)^3 d\theta$, sans être un modèle de base, il est commode de retenir cette intégrale, ce qui nous évite de devoir l'obtenir encore et toujours en intégrant pas parties.

```
> value((63));
```

$$= \frac{27}{2} \sec(\theta) \tan(\theta) + \frac{19}{2} \ln(\sec(\theta) + \tan(\theta)) \quad (64)$$

Les résultats ne correspondent pas à ceux retenus en classe dûs au mécanisme de la simplification automatique (MSAM). Corrigeons la situation.

```
> op(1,op(2,(64)));
op(1,op(2,op(2,op(2,(64)))));
```

$$\frac{\frac{27}{2} \sec(\theta) \tan(\theta)}{\sec(\theta) + \tan(\theta)} \quad (65)$$

$$\begin{aligned} > \text{subs}(\text{op}(1, \text{op}(2, (64))) = \text{abs}(\text{op}(1, \text{op}(2, (64))))), (64)); \\ &= \frac{27}{2} |\sec(\theta) \tan(\theta)| + \frac{19}{2} \ln(\sec(\theta) + \tan(\theta)) \end{aligned} \quad (66)$$

$$\begin{aligned} > \text{subs}(\text{op}(1, \text{op}(2, \text{op}(2, \text{op}(2, (64)))))) = \text{abs}(\text{op}(1, \text{op}(2, \text{op}(2, \text{op}(2, (64)))))), \\ (66)); \\ &= \frac{27}{2} |\sec(\theta) \tan(\theta)| + \frac{19}{2} \ln(|\sec(\theta) + \tan(\theta)|) \end{aligned} \quad (67)$$

Effectuons la substitution à rebours $\theta = \arccos\left(\frac{3}{x}\right)$.

$$\begin{aligned} > \text{eval}((67), \text{theta} = \arccos(3/x)); \\ &= \frac{3}{2} |x|^2 \sqrt{\left|1 - \frac{9}{x^2}\right|} + \frac{19}{2} \ln\left(\frac{1}{3} \left|x + \sqrt{1 - \frac{9}{x^2}} x\right|\right) \end{aligned} \quad (68)$$

$$\begin{aligned} > \text{``} = \text{simplify}(\text{rhs}((68))) \text{ assuming } x > 3; \\ &= \frac{3}{2} x \sqrt{x^2 - 9} - \frac{19}{2} \ln(3) + \frac{19}{2} \ln(x + \sqrt{x^2 - 9}) \end{aligned} \quad (69)$$

$$\begin{aligned} > \text{Rép_finale_1} := \text{Problème} = \text{rhs}((69)) + 19 * \ln(3) * (1/2) + C1; \\ \text{Rép_finale_1} := \int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx = \frac{3}{2} x \sqrt{x^2 - 9} + \frac{19}{2} \ln(x + \sqrt{x^2 - 9}) + C1 \end{aligned} \quad (70)$$

Simplification conditionnelle pour $\theta \in \left]0, \frac{\pi}{2}\right[$.

Trouvons Intégrale_2.

$$\begin{aligned} > \text{Intégrale_2} = \text{expand}(\text{Intégrale_2}); \\ &\int \frac{4 \cos(\theta)^2 - 27}{\cos(\theta)^3} d\theta = \int \left(\frac{4}{\cos(\theta)} - \frac{27}{\cos(\theta)^3} \right) d\theta \end{aligned} \quad (71)$$

$$\begin{aligned} > \text{``} = \text{algsubs}(\cos(\text{theta})^{(-1)} = \sec(\text{theta}), \text{rhs}((71))); \\ &= \int (-27 \sec(\theta)^3 + 4 \sec(\theta)) d\theta \end{aligned} \quad (72)$$

$$> \text{``} = \text{expand}(\text{rhs}((72))); \quad (73)$$

$$= -27 \left(\int \sec(\theta)^3 d\theta \right) + 4 \left(\int \sec(\theta) d\theta \right) \quad (73)$$

On reconnaît le modèle $\int \sec(\theta) d\theta$, par contre, $\int \sec(\theta)^3 d\theta$, sans être un modèle de base, il est commode de retenir cette intégrale, ce qui nous évite de devoir l'obtenir encore et toujours en intégrant pas parties.

Cette fois, posons directement la réponse.

$$\begin{aligned} > \text{``} = -27 * \text{abs}(\sec(\theta) * \tan(\theta)) * (1/2) - 19 * \ln(\text{abs}(\sec(\theta) + \tan(\theta))) * (1/2) \\ &= -\frac{27}{2} |\sec(\theta) \tan(\theta)| - \frac{19}{2} \ln(|\sec(\theta) + \tan(\theta)|) \end{aligned} \quad (74)$$

Effectuons la substitution à rebours $\theta = \arccos\left(\frac{3}{x}\right)$.

$$\begin{aligned} > \text{eval}((74), \theta = \arccos(3/x)); \\ &= -\frac{3}{2} |x|^2 \sqrt{\left|1 - \frac{9}{x^2}\right|} - \frac{19}{2} \ln\left(\frac{1}{3} \left|x + \sqrt{1 - \frac{9}{x^2}} x\right|\right) \end{aligned} \quad (75)$$

$$\begin{aligned} > \text{``} = \text{simplify}(\text{rhs}((75))) \text{ assuming } x < -3; \\ &= \frac{3}{2} x \sqrt{x^2 - 9} + \frac{19}{2} \ln(3) - \frac{19}{2} \ln(-x + \sqrt{x^2 - 9}) \end{aligned} \quad (76)$$

$$\begin{aligned} > \text{Rép_finale_2} := \text{Problème} = \text{rhs}((76)) - 19 * \ln(3) * (1/2) + C2; \\ \text{Rép_finale_2} := \int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx = \frac{3}{2} x \sqrt{x^2 - 9} - \frac{19}{2} \ln(-x + \sqrt{x^2 - 9}) + C2 \end{aligned} \quad (77)$$

Résumons.

Pour $x > 3$:

$$\begin{aligned} > \text{Rép_finale_1}; \\ \int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx = \frac{3}{2} x \sqrt{x^2 - 9} + \frac{19}{2} \ln(x + \sqrt{x^2 - 9}) + C1 \end{aligned} \quad (78)$$

Pour $x < -3$:

$$\begin{aligned} > \text{Rép_finale_2}; \\ \int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx = \frac{3}{2} x \sqrt{x^2 - 9} - \frac{19}{2} \ln(-x + \sqrt{x^2 - 9}) + C2 \end{aligned} \quad (79)$$

Vérification.

$$\begin{aligned} > \text{Diff}(\text{rhs}(\text{Rép_finale_1}), x) = \text{diff}(\text{rhs}(\text{Rép_finale_1}), x); \\ \frac{\partial}{\partial x} \left(\frac{3}{2} x \sqrt{x^2 - 9} + \frac{19}{2} \ln(x + \sqrt{x^2 - 9}) + C1 \right) = \frac{3}{2} \sqrt{x^2 - 9} + \frac{3}{2} \frac{x^2}{\sqrt{x^2 - 9}} \end{aligned} \quad (80)$$

$$+ \frac{19}{2} \frac{1 + \frac{x}{\sqrt{x^2-9}}}{x + \sqrt{x^2-9}}$$

> simplify((80));

$$\frac{\partial}{\partial x} \left(\frac{3}{2} x \sqrt{x^2-9} + \frac{19}{2} \ln(x + \sqrt{x^2-9}) + C1 \right) = \frac{3x^2-4}{\sqrt{x^2-9}} \quad (81)$$

> Diff(rhs(Rép_finale_2),x)=diff(rhs(Rép_finale_2),x);

$$\frac{\partial}{\partial x} \left(\frac{3}{2} x \sqrt{x^2-9} - \frac{19}{2} \ln(-x + \sqrt{x^2-9}) + C2 \right) = \frac{3}{2} \sqrt{x^2-9} + \frac{3}{2} \frac{x^2}{\sqrt{x^2-9}} \quad (82)$$

$$- \frac{19}{2} \frac{-1 + \frac{x}{\sqrt{x^2-9}}}{-x + \sqrt{x^2-9}}$$

> simplify((82));

$$\frac{\partial}{\partial x} \left(\frac{3}{2} x \sqrt{x^2-9} - \frac{19}{2} \ln(-x + \sqrt{x^2-9}) + C2 \right) = \frac{3x^2-4}{\sqrt{x^2-9}} \quad (83)$$

Ce qui donne bien dans les deux cas l'intégrande.

> Integrand(Problème)[1];

$$\frac{3x^2-4}{\sqrt{x^2-9}} \quad (84)$$

Conclusion:

$$\text{Pour } x < -3, \int \frac{3x^2-4}{\sqrt{x^2-9}} dx = -\frac{19 \ln(-x + \sqrt{x^2-9})}{2} + \frac{3\sqrt{x^2-9} x}{2} + C1$$

$$\text{Pour } x > 3, \int \frac{3x^2-4}{\sqrt{x^2-9}} dx = \frac{19 \ln(x + \sqrt{x^2-9})}{2} + \frac{3\sqrt{x^2-9} x}{2} + C2$$

Discussion

Le logiciel via *Student[Calculus I][IntTutor]* propose cette substitution algébrique:

> with(Student[Calculus1], IntTutor);
[IntTutor] (85)

Exécuter avec «*Toutes les étapes*» puis «*Fermer*». Tout le développement sera transposé dans votre document.

> IntTutor(Int((3*x^2-4)/sqrt(x^2-9), x));

$$\begin{aligned}
 & \int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx \\
 &= \int \left(-\frac{3u}{4} + \frac{-38u^2 - 243}{4u^3} \right) du \\
 &= \int -\frac{3u}{4} du + \int \frac{-38u^2 - 243}{4u^3} du \\
 &= -\frac{3 \left(\int u du \right)}{4} + \int \frac{-38u^2 - 243}{4u^3} du \\
 &= -\frac{3u^2}{8} + \int \frac{-38u^2 - 243}{4u^3} du \\
 &= -\frac{3u^2}{8} + \frac{\left(\int \frac{-38u^2 - 243}{u^3} du \right)}{4} \\
 &= -\frac{3u^2}{8} + \frac{\left(\int \left(-\frac{38}{u} - \frac{243}{u^3} \right) du \right)}{4} \\
 &= -\frac{3u^2}{8} + \frac{\left(\int -\frac{38}{u} du \right)}{4} + \frac{\left(\int -\frac{243}{u^3} du \right)}{4} \\
 &= -\frac{3u^2}{8} - \frac{19 \left(\int \frac{1}{u} du \right)}{2} + \frac{\left(\int -\frac{243}{u^3} du \right)}{4} \\
 &= -\frac{3u^2}{8} - \frac{19 \ln(u)}{2} + \frac{\left(\int -\frac{243}{u^3} du \right)}{4} \\
 &= -\frac{3u^2}{8} - \frac{19 \ln(u)}{2} - \frac{243 \left(\int \frac{1}{u^3} du \right)}{4} \\
 &= -\frac{3u^2}{8} - \frac{19 \ln(u)}{2} + \frac{243}{8u^2} \\
 &= -\frac{3x^2}{4} + \frac{3x\sqrt{x^2-9}}{4} + \frac{27}{8} - \frac{19 \ln(-x + \sqrt{x^2-9})}{2} + \frac{243}{8(-x + \sqrt{x^2-9})^2}
 \end{aligned}$$

$$\int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx = -\frac{3}{4} x^2 + \frac{3}{4} x\sqrt{x^2 - 9} + \frac{27}{8} - \frac{19}{2} \ln(-x + \sqrt{x^2 - 9}) \quad (86)$$

$$+ \frac{243}{8 \left(-x + \sqrt{x^2 - 9} \right)^2}$$

Pouvez-vous affirmer que cette primitive est acceptable sur tout le domaine de $\int \frac{3x^2 - 4}{\sqrt{x^2 - 9}} dx$? (Évidemment

non ! Seulement pour $x < -3$)

> Diff(rhs((86)),x)=diff(rhs((86)),x)

$$\frac{d}{dx} \left(-\frac{3}{4}x^2 + \frac{3}{4}x\sqrt{x^2-9} + \frac{27}{8} - \frac{19}{2}\ln(-x + \sqrt{x^2-9}) + \frac{243}{8(-x + \sqrt{x^2-9})^2} \right) \quad (87)$$

$$= -\frac{3}{2}x + \frac{3}{4}\sqrt{x^2-9} + \frac{3}{4}\frac{x^2}{\sqrt{x^2-9}} - \frac{19}{2}\frac{-1 + \frac{x}{\sqrt{x^2-9}}}{-x + \sqrt{x^2-9}}$$

$$- \frac{243}{4}\frac{-1 + \frac{x}{\sqrt{x^2-9}}}{(-x + \sqrt{x^2-9})^3}$$

> ``=simplify(rhs((87)),radical);

$$= -\frac{(3x^2 - 4)(2x\sqrt{x^2 - 9} - 2x^2 + 9)}{\sqrt{x^2 - 9}(-x + \sqrt{x^2 - 9})^2} \quad (88)$$

> ``=rationalize(rhs((88)));

$$= -\frac{1}{81}\frac{(2x\sqrt{x^2-9} - 2x^2 + 9)(2\sqrt{x^2-9}x^2 + 2x^3 - 9\sqrt{x^2-9} - 18x)(3x^2 - 4)}{x^2 - 9} \quad (89)$$

> expand((89));

$$= \frac{3x^2}{\sqrt{x^2-9}} - \frac{4}{\sqrt{x^2-9}} \quad (90)$$

> normal((90));

$$= \frac{3x^2 - 4}{\sqrt{x^2 - 9}} \quad (91)$$

Exemple 4

Calculons l'intégrale indéfinie $\int \frac{\sqrt{9-x^2}}{2x} dx$

Avec Maple directement:

```
> f:=x->sqrt(9-x^2)/(2*x);
```

$$f := x \rightarrow \frac{1}{2} \frac{\sqrt{9-x^2}}{x} \quad (92)$$

Posons l'intégrale indéfinie calculer.

```
> Problème:=Int(f(x),x);
```

$$\text{Problème} := \int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx \quad (93)$$

Intégrons directement avec la macro-commande int.

```
> Rép_Maple:=Problème=value(Problème)+C;
```

$$\text{Rép_Maple} := \int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = \frac{1}{2} \sqrt{-x^2+9} - \frac{3}{2} \operatorname{arctanh}\left(\frac{3}{\sqrt{-x^2+9}}\right) + C \quad (94)$$

Remarque: Le résultat donné est exprimé avec la fonction $\operatorname{arctanh}$ (Arc tangente hyperbolique). L'apparition de ce modèle de fonction est un tantinet surprenante pour le niveau du cours NYB.

Avec un développement pas à pas jusqu'aux formules de base.

```
> Problème;
```

$$\int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx \quad (95)$$

Posons $x = 3 \sin(\theta)$ où $\theta \in \left[-\frac{\pi}{2}, \frac{\pi}{2}\right] \setminus \{0\}$.

```
> Rule[change,x=3*sin(theta),theta]((95));;
```

```
Creating problem #4
```

```
Applying substitution x = 3*sin(theta), theta = arcsin(1/3*x) with dx = 3*cos(theta)*dtheta, dtheta = 1/(-x^2+9)^(1/2)*dx
```

$$\int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = \int \left(-\frac{3}{2} \sin(\theta) + \frac{3}{2 \sin(\theta)} \right) d\theta \quad (96)$$

```
> Rule[rewrite,1/sin(theta)=csc(theta)]((96));
```

$$\int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = \int \left(-\frac{3}{2} \sin(\theta) + \frac{3}{2} \csc(\theta) \right) d\theta \quad (97)$$

```
> Rule[sum]((97));
```

$$\int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = \int \left(-\frac{3}{2} \sin(\theta) \right) d\theta + \int \frac{3}{2} \csc(\theta) d\theta \quad (98)$$

```
> Rule[`c*`]((98));
```

$$\int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = -\frac{3}{2} \int \sin(\theta) d\theta + \int \frac{3}{2} \csc(\theta) d\theta \quad (99)$$

> Rule[`c*`](99);

$$\int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = -\frac{3}{2} \int \sin(\theta) d\theta + \frac{3}{2} \int \csc(\theta) d\theta \quad (100)$$

Ons reconnaît deux modèle de base.

> Rule[sin](100);

$$\int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = \frac{3}{2} \cos(\theta) + \frac{3}{2} \int \csc(\theta) d\theta \quad (101)$$

> Rule[csc](101);

$$\int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = \frac{3}{2} \cos(\theta) - \frac{3}{2} \ln(\csc(\theta) + \cot(\theta)) \quad (102)$$

Ce résultat ne correspond pas au modèle $\int \csc(u) u'$ retenu en classe. Ajustons cette réponse.

> op(1,op(2,op(2,op(2,(102)))));

$$\csc(\theta) + \cot(\theta) \quad (103)$$

> algsubs(op(1,op(2,op(2,op(2,(102)))))=abs(op(1,op(2,op(2,op(2,(102))))), (102));

$$\int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = \frac{3}{2} \cos(\theta) - \frac{3}{2} \ln(|\csc(\theta) + \cot(\theta)|) \quad (104)$$

Ayant posé $x = 3 \sin(\theta)$, substituons à rebours $\theta = \arcsin\left(\frac{x}{3}\right)$.

> ``=subs(theta=arcsin(x/3),rhs(104));

$$= \frac{3}{2} \cos\left(\arcsin\left(\frac{1}{3}x\right)\right) - \frac{3}{2} \ln\left(\left|\csc\left(\arcsin\left(\frac{1}{3}x\right)\right) + \cot\left(\arcsin\left(\frac{1}{3}x\right)\right)\right|\right) \quad (105)$$

> simplify(105);

$$= \frac{1}{2} \sqrt{-x^2+9} - \frac{3}{2} \ln\left(\left|\frac{3+\sqrt{-x^2+9}}{x}\right|\right) \quad (106)$$

> Rép_finale:=Problème=rhs(106)+C;

$$\text{Rép_finale} := \int \frac{1}{2} \frac{\sqrt{-x^2+9}}{x} dx = \frac{1}{2} \sqrt{-x^2+9} - \frac{3}{2} \ln\left(\left|\frac{3+\sqrt{-x^2+9}}{x}\right|\right) + C \quad (107)$$

Vérification conditionnelle sur le domaine de l'intégrale.

$$\begin{aligned} &> \text{Rép_simplifiée_1} := \text{simplify}(\text{rhs}((107))) \text{ assuming } x > 0, x \leq 3; \\ &\text{Rép_simplifiée_1} := \frac{1}{2} \sqrt{-x^2 + 9} + \frac{3}{2} \ln(x) - \frac{3}{2} \ln(3 + \sqrt{-x^2 + 9}) + C \end{aligned} \quad (108)$$

$$\begin{aligned} &> \text{Diff}(\text{Rép_simplifiée_1}, x) = \text{diff}(\text{Rép_simplifiée_1}, x); \\ &\frac{\partial}{\partial x} \left(\frac{1}{2} \sqrt{-x^2 + 9} + \frac{3}{2} \ln(x) - \frac{3}{2} \ln(3 + \sqrt{-x^2 + 9}) + C \right) = -\frac{1}{2} \frac{x}{\sqrt{-x^2 + 9}} + \frac{3}{2x} \\ &+ \frac{3}{2} \frac{x}{\sqrt{-x^2 + 9} (3 + \sqrt{-x^2 + 9})} \end{aligned} \quad (109)$$

$$\begin{aligned} &> \text{``} = \text{rationalize}(\text{rhs}((109))); \\ &= -\frac{1}{2} \frac{(-x^2 + 3\sqrt{-x^2 + 9} + 9) (\sqrt{-x^2 + 9} - 3)}{x^3} \end{aligned} \quad (110)$$

$$\begin{aligned} &> \text{``} = \text{expand}(\text{rhs}((110))); \\ &= \frac{1}{2} \frac{\sqrt{-x^2 + 9}}{x} \end{aligned} \quad (111)$$

$$\begin{aligned} &> \text{Rép_simplifiée_2} := \text{simplify}(\text{rhs}((107))) \text{ assuming } x \geq -3, x < 0; \\ &\text{Rép_simplifiée_2} := \frac{1}{2} \sqrt{-x^2 + 9} + \frac{3}{2} \ln(-x) - \frac{3}{2} \ln(3 + \sqrt{-x^2 + 9}) + C \end{aligned} \quad (112)$$

$$\begin{aligned} &> \text{Diff}(\text{Rép_simplifiée_2}, x) = \text{diff}(\text{Rép_simplifiée_2}, x); \\ &\frac{\partial}{\partial x} \left(\frac{1}{2} \sqrt{-x^2 + 9} + \frac{3}{2} \ln(-x) - \frac{3}{2} \ln(3 + \sqrt{-x^2 + 9}) + C \right) = -\frac{1}{2} \frac{x}{\sqrt{-x^2 + 9}} + \frac{3}{2x} \\ &+ \frac{3}{2} \frac{x}{\sqrt{-x^2 + 9} (3 + \sqrt{-x^2 + 9})} \end{aligned} \quad (113)$$

$$\begin{aligned} &> \text{``} = \text{rationalize}(\text{rhs}((113))); \\ &= -\frac{1}{2} \frac{(-x^2 + 3\sqrt{-x^2 + 9} + 9) (\sqrt{-x^2 + 9} - 3)}{x^3} \end{aligned} \quad (114)$$

$$\begin{aligned} &> \text{``} = \text{expand}(\text{rhs}((114))); \\ &= \frac{1}{2} \frac{\sqrt{-x^2 + 9}}{x} \end{aligned} \quad (115)$$

Ce qui, dans les deux cas, est bien l'intégrande de Problème.

> Integrand(Problème)[1];

$$\frac{1}{2} \frac{\sqrt{-x^2 + 9}}{x} \quad (116)$$

Exemple 5

Calculons l'intégrale indéfinie $\int \frac{1}{\sqrt{x^2 - 4x + 13}} dx$.

Avec Maple directement:

```
> f:=x->1/sqrt(x^2-4*x+13);
```

$$f := x \rightarrow \frac{1}{\sqrt{x^2 - 4x + 13}} \quad (117)$$

Posons l'intégrale indéfinie en lui donnant le nom Problème.

```
> Problème:=Int(f(x),x);
```

$$\text{Problème} := \int \frac{1}{\sqrt{x^2 - 4x + 13}} dx \quad (118)$$

Intégrons directement avec la macro-commande int.

```
> Rép_Maple:=Problème=value(Problème)+C;
```

$$\text{Rép_Maple} := \int \frac{1}{\sqrt{x^2 - 4x + 13}} dx = \operatorname{arcsinh}\left(\frac{1}{3}x - \frac{2}{3}\right) + C \quad (119)$$

Remarque: Le résultat donné est exprimé avec la fonction $\operatorname{arcsinh}$ (Arc sinus hyperbolique). L'apparition de ce modèle de fonction est un tantinet surprenante pour le cours NYB.

Maintenant avec un développement pas à pas jusqu'aux formules de base.

```
> Problème;
```

$$\int \frac{1}{\sqrt{x^2 - 4x + 13}} dx \quad (120)$$

Complétons le carré de $x^2 - 4x + 13$. Faisons appel directement à la macro-commande `CompleteSquare` de la bibliothèque `Student[Precalculus]`.

```
> Student[Precalculus][CompleteSquare]((120),x);
```

$$\int \frac{1}{\sqrt{(x-2)^2 + 9}} dx \quad (121)$$

Intégrons par substitution trigonométrique. Posons $x - 2 = 3 \tan(\theta)$ où $\theta \in \left] -\frac{\pi}{2}, \frac{\pi}{2} \right[$.

```
> Rule[change,x-2=3*tan(theta),theta]((121));
Creating problem #5
```

Applying substitution $x = 3 \cdot \tan(\theta) + 2$, $\theta = \arctan(1/3 \cdot x - 2/3)$ with $dx = (3 + 3 \cdot \tan(\theta)^2) \cdot d\theta$, $d\theta = 1/3 / ((1/3 \cdot x - 2/3)^2 + 1) \cdot dx$

$$\int \frac{1}{\sqrt{(x-2)^2 + 9}} dx = \int \sec(\theta) d\theta \quad (122)$$

On reconnaît le modèle $\int \sec(\theta) d\theta$.

> **Rule[sec]((122));**

$$\int \frac{1}{\sqrt{(x-2)^2 + 9}} dx = \ln(\sec(\theta) + \tan(\theta)) \quad (123)$$

Encore une fois, ajustons la réponse précédente en conformité avec le modèle retenu en classe.

> **Problème=ln(abs(sec(theta)+tan(theta)));**

$$\int \frac{1}{\sqrt{x^2 - 4x + 13}} dx = \ln(|\sec(\theta) + \tan(\theta)|) \quad (124)$$

Effectuons la substitution à rebours.

> **subs(theta=arctan((x-2)/3),(124));**

$$\int \frac{1}{\sqrt{x^2 - 4x + 13}} dx = \ln\left(\left|\sec\left(\arctan\left(\frac{1}{3}x - \frac{2}{3}\right)\right) + \tan\left(\arctan\left(\frac{1}{3}x - \frac{2}{3}\right)\right)\right|\right) \quad (125)$$

> **simplify((125));**

$$\int \frac{1}{\sqrt{x^2 - 4x + 13}} dx = -\ln(3) + \ln\left(\left|\sqrt{x^2 - 4x + 13} + x - 2\right|\right) \quad (126)$$

> **Rép:=(126)+(0=ln(3)+C);**

$$\text{Rép} := \int \frac{1}{\sqrt{x^2 - 4x + 13}} dx = \ln\left(\left|\sqrt{x^2 - 4x + 13} + x - 2\right|\right) + C \quad (127)$$

Puisque $\frac{\sqrt{x^2 - 4x + 13}}{3} + \frac{x}{3} - \frac{2}{3} > 0$, (pourquoi?), on simplifie l'argument du logarithme comme

suit:

> **Rép_finale:=Problème=ln(sqrt(x^2-4*x+13)+x-2)+C;**

$$\text{Rép_finale} := \int \frac{1}{\sqrt{x^2 - 4x + 13}} dx = \ln\left(\sqrt{x^2 - 4x + 13} + x - 2\right) + C \quad (128)$$

Vérification:

> **Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);**

$$\frac{\partial}{\partial x} \left(\ln(\sqrt{x^2 - 4x + 13} + x - 2) + C \right) = \frac{\frac{1}{2} \frac{2x - 4}{\sqrt{x^2 - 4x + 13}} + 1}{\sqrt{x^2 - 4x + 13} + x - 2} \quad (129)$$

```
> ``=normal(rhs((129)));
```

$$= \frac{1}{\sqrt{x^2 - 4x + 13}} \quad (130)$$

La simplification de la dérivée correspond bien à l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$\frac{1}{\sqrt{x^2 - 4x + 13}} \quad (131)$$